NAG Fortran Library Routine Document

F07CEF (DGTTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CEF (DGTTRS) computes the solution to a real system of linear equations $AX = B$ or $A^{\mathrm{T}}X = B$, where A is an n by n tridiagonal matrix and X and B are n by r matrices, using the LU factorization returned by F07CDF (DGTTRF).

2 Specification

```
SUBROUTINE F07CEF (TRANS, N, NRHS, DL, D, DU, DU2, IPIV, B, LDB, INFO)

INTEGER             N, NRHS, IPIV(*), LDB, INFO
double precision    DL(*), D(*), DU(*), DU2(*), B(LDB,*)
CHARACTER*1         TRANS
```

The routine may be called by its LAPACK name ***dgttrs***.

3 Description

F07CEF (DGTTRS) should be preceded by a call to F07CDF (DGTTRF), which uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one non-zero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals. F07CEF (DGTTRS) then utilizes the factorization to solve the required equations.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

5 Parameters

1: TRANS – CHARACTER*1 *Input*

On entry: specifies the equations to be solved as follows:

TRANS = 'N'

Solve $AX = B$ for X.

TRANS = 'T' or 'C'

Solve $A^{\mathrm{T}}X = B$ for X.

Constraint: TRANS = 'N', 'T' or 'C'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS ≥ 0.

4: DL($*$) – ***double precision*** array *Input*

Note: the dimension of the array DL must be at least $\max(1, \mathrm{N}-1)$.

On entry: must contain the $(n-1)$ multipliers that define the matrix L of the LU factorization of A.

5: D($*$) – ***double precision*** array *Input*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the upper triangular matrix U from the LU factorization of A.

6: DU($*$) – ***double precision*** array *Input*

Note: the dimension of the array DU must be at least $\max(1, \mathrm{N}-1)$.

On entry: must contain the $(n-1)$ elements of the first superdiagonal of U.

7: DU2($*$) – ***double precision*** array *Input*

Note: the dimension of the array DU2 must be at least $\max(1, \mathrm{N}-2)$.

On entry: must contain the $(n-2)$ elements of the second superdiagonal of U.

8: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n pivot indices that define the permutation matrix P. At the ith step, row i of the matrix was interchanged with row IPIV(i), and IPIV(i) must always be either i or $(i+1)$, IPIV(i) $= i$ indicating that a row interchange was not performed.

9: B(LDB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r matrix of right-hand sides B.

On exit: the n by r solution matrix X.

10: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07CEF (DGTTRS) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

11: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\|A^{-1}\right\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

Following the use of this routine F07CGF (DGTCON) can be used to estimate the condition number of A and F07CHF (DGTRFS) can be used to obtain approximate error bounds.

8 Further Comments

The total number of floating-point operations required to solve the equations $AX = B$ or $A^{\mathrm{T}}X = B$ is proportional to $n \times r$.

The complex analogue of this routine is F07CSF (ZGTTRS).

9 Example

This example solves the equations

$$AX = B,$$

where A is the tridiagonal matrix

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 2.7 & 6.6 \\ -0.5 & 10.8 \\ 2.6 & -3.2 \\ 0.6 & -11.2 \\ 2.7 & 19.1 \end{pmatrix}.$$

9.1 Program Text

```
*     F07CEF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHSMX
      PARAMETER        (NMAX=50,NRHSMX=4)
      INTEGER          LDB
      PARAMETER        (LDB=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N, NRHS
*     .. Local Arrays ..
      DOUBLE PRECISION B(LDB,NRHSMX), D(NMAX), DL(NMAX-1), DU(NMAX-1),
     +                 DU2(NMAX-2)
      INTEGER          IPIV(NMAX)
*     .. External Subroutines ..
      EXTERNAL         DGTTRF, DGTTRS, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07CEF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHSMX) THEN
```

```
*
*        Read the tridiagonal matrix A from data file
*
         READ (NIN,*) (DU(I),I=1,N-1)
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (DL(I),I=1,N-1)
*
*        Read the right hand matrix B
*
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Factorize the tridiagonal matrix A
*
         CALL DGTTRF(N,DL,D,DU,DU2,IPIV,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Solve the equations AX = B
*
            CALL DGTTRS('No transpose',N,NRHS,DL,D,DU,DU2,IPIV,B,LDB,
     +                  INFO)
*
*           Print the solution
*
            IFAIL = 0
            CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
*
         ELSE
            WRITE (NOUT,99999) 'The (', INFO, ',', INFO, ')',
     +        ' element of the factor U is zero'
         END IF
       ELSE
          WRITE (NOUT,*) 'NMAX and/or NRHSMX too small'
       END IF
       STOP
*
99999 FORMAT (1X,A,I3,A,I3,A,A)
      END
```

9.2 Program Data

```
F07CEF Example Program Data
  5     2                        :Values of N and NRHS
        2.1  -1.0   1.9   8.0
  3.0   2.3  -5.0  -0.9   7.1
  3.4   3.6   7.0  -6.0         :End of matrix A
  2.7   6.6
 -0.5  10.8
  2.6  -3.2
  0.6 -11.2
  2.7  19.1                     :End of matrix B
```

9.3 Program Results

```
 F07CEF Example Program Results

 Solution(s)
              1          2
 1     -4.0000     5.0000
 2      7.0000    -4.0000
 3      3.0000    -3.0000
 4     -4.0000    -2.0000
 5     -3.0000     1.0000
```